

# FORREST M. MIMS III THE COMPUTER SCIENTIST

## OPTOELECTRONIC DIGITIZER




Fig. 1. How a light sensor can be installed in a plotter pen.

**D**IGITIZERS permit information ranging from simple data to complex shapes and patterns to be manually entered into a computer. Many kinds of low-cost digitizing devices are available for personal computers, including joysticks, mice, trackballs, and graphics tablets. Even some *xy* plotters can be adapted to function as digitizers. Automatic digitizers are available, but they are quite expensive.

Recently I've been experimenting with various ways to convert an *xy* plotter into an automatic digitizer. The results thus far have been very encouraging, and my present set-up, which I will describe here, can be programmed to grade multiple-choice tests, tally surveys, and digitize low-resolution shapes and images, all automatically. If you already own or have access to a computer and compatible *xy* plotter, the cost for this powerful capability is surprisingly low.

### Optical Digitizing Methods

The automatic digitizing methods with which I've experimented are all based upon the optoelectronic (or photonic) sensing of the presence or absence of light or dark markings on ordinary paper. In operation, a computer is programmed to sweep the pen carriage of an *xy* plotter across a paper containing information or an image to be digitized. An optoelectronic sensor installed in the pen carriage is connected to the computer's joystick port. A simple software routine permits the signal level from the sensor to be correlated with the precise location of its origin.

Although many different optical sensing methods are available, the simplest is to install a light-sensitive detector in a

discarded plotter pen as shown in Fig. 1. The sensitive surface of the detector can then view the paper through the narrow aperture through which the ink once passed. If connections to the sensor are made with flexible wrapping wire, a modified pen containing a detector can be automatically returned to and retrieved from a pen stall.

Though this method will work, it requires careful attention to lighting. The paper under the moving sensor must be evenly illuminated, or erroneous signals will result. One solution to this problem is to substitute a reflective optocoupler for the detector. This device contains both a light emitting diode and a phototransistor. Since this sensor includes a built-in light source, it is not dependent upon external illumination.

Unfortunately, most reflective optocouplers are too large to fit inside a plotter pen. Some that might, however, are made by Skan-A-Matic Corporation (PO Box S, Route 5W, Elbridge, NY 13060).

Another sensing method that works

quite well is to couple both the optical detector and a source of illumination to the plotter's pen carriage by means of fiber optics. This method permits very reliable, low-resolution digitizing with a minimum of external hardware. Indeed, when used with a *PCjr*, this method works exceptionally well with a readily variable cadmium sulfide photocell as a light detector. The photocell can be directly connected to one of the joystick ports on the *PCjr*.

The method can also be used with a Color Computer and certain other computers having joystick ports simply by adding a series resistor to the photocell and connecting the two components across 5 V. The resulting circuit provides an output voltage that varies in amplitude according to the light level on the photocell.

In short, it's possible to assemble an automatic optical digitizer from many different combinations of computers, plotters, and sensors. Following is a detailed description of one of the many possible arrangements. You may not happen to have the equipment I used, but with some planning you should be able to adapt the principles described to your particular system.

### An Automatic Digitizer

The hardware necessary to transform an *xy* plotter connected to a *PCjr* into an automatic fiber optic digitizer is shown in Fig. 2. The sensor is a cadmium sulfide photoresistor having a low light resistance and a high dark resistance. A suitable choice is Radio Shack's Catalog





Fig. 2. Fiber optic automatic digitizer.

No. 276-116.

Since cadmium sulfide photocells have a relatively slow response time, illumination can be provided by either a battery or line-powered lamp. The photocell will ignore the 60-Hz intensity fluctuations from the line-powered lamp.

The fiber optic cable is the most important, and most expensive, component in the system. The *bifurcated* fiber optic cable used was formed from two cables merged into one in a Y configuration. The fibers in the two cables are randomly mixed in the stem of the Y.


**Fig. 3. A bifurcated fiber optic cable with light sensor and penlight.**



Light is injected into one branch of the Y where it evenly illuminates as pot directly below the end of the merged cables. Some of the light reflected from the paper enters the end of the merged fiber cable. Half this light travels up the side of the Y terminated by the photosensor.

If you've never worked with fiber optics, you may wonder how effective a bifurcated fiber optic cable is in this role. I found it to be exceptionally effective, and it is easy to observe the change in light intensity at the photosensor end of the cable when the sensing end of the Y cable passes over a black line drawn on white paper.

Bifurcated fiber optic cables are available from several companies, or you can make your own. The one I used is Dolan-Jenner's Type No. EE824. It is 24" long and costs \$60.00. Before ordering a cable, you may wish to request literature about the other bifurcated fiber optic cables made by the same company. Send your inquiry to Dolan-Jenner Industries, PO Box 1020, Woburn, MA 01801.



**Fig. 5 PCjr joystick connections.**

Some of this company's cables have a smaller diameter and can be used to provide higher resolution. Keep in mind Dolan-Jenner has a \$50 minimum order requirement.

### Assembling the Digitizer

There are many ways to attach the photocell to the terminal at the end of one of the two branches of the bifurcated cable. One way is to insert the photocell into a length of heat-shrinkable tubing. Wrap the metal terminal on one end of the fiber optic cable with tape to increase its diameter and insert the wrapped end into an aluminum bushing previously inserted into the heat-shrinkable tubing.

The other end of the Y should be coupled to a light source. This can be done

by inserting the end of the cable, wrapped with tape to increase its diameter, into a plastic or metal cylinder placed over the end of a small two-cell penlight. Another is to clip the free end of the cable to a small desk lamp with a small clamp or alligator clip.


Figure 3 is a photograph of the bifurcated fiber optic cable used in my experiments. A flashlight and detector have been coupled to the ends of the Y, and the stem of the cable is ready to be inserted into the plotter's pen carriage.

The terminal at the end of the stem of the Y should be increased in diameter with tape or heat-shrinkable tubing and then *carefully* inserted into the top of the pen holder of a plotter. Although I used a Hewlett-Packard HP7470 plotter, others may also work.

After the fiber optic cable has been arranged, the photocell can be connected to the joystick port of a PCjr. Since it's difficult to find plugs that fit Junior's two joystick sockets, you may want to do as I have and modify a joystick by adding a small phone jack to it. Connect the terminals of the jack across the x-axis potentiometer. If the jack is the kind designed to switch a speaker off when a phone plug is inserted, use the switch mechanism to disconnect the potentiometer when a plug connected to the photocell is inserted. Insulated alligator clips soldered to a pair of wires connected to the plug can be clipped to the photocell's leads.

Alternatively, you can connect the


**Fig. 4. Internal circuitry of a PCjr joystick.**



## Computer Scientist

photocell directly to Junior's joystick connector by using a hand-held wire-wrap tool. The wires can be removed later.

Unless they are shielded, the two wires connecting the photocell to the joystick jack or Junior's joystick socket should be relatively short. Otherwise, external electrical noise may be coupled




**Fig. 6. Photograph of the author's automatic digitizer system.**

into the computer and cause erroneous joystick readings.

In any event, always use caution when making modifications to computer equipment. A manufacturer may consider the machine's warranty voided should your modifications cause components to be damaged.

Figure 4 shows the internal circuitry of Junior's joysticks. The STICK(O) command addresses the x-axis potentiometer, the one for which the photocell should be substituted. Figure 5 shows the connections between Junior and a joystick and also identifies the pin posi-



**Fig. 8. Typical shapes scanned by automatic digitizer.**

tions of the joystick's socket.

After the fiber optic sensing apparatus is completed, it's time to connect the plotter to the PCjr. The RS-232 version of the HP7470 should be connected to the serial ("S") port on the back of the Junior through a null modem connector. Depending upon the cable used, you may also need a female-to-female adapter.

Next, it's necessary to make sure Junior and the plotter are on speaking terms. If you use the HP7470, check the settings on the DIP switch adjacent to the RS-232 connector. Locations US and B4 should be set at the 1 position. All other locations should be set at the 0 position.

If you use the same equipment I did, your system should now resemble the setup shown in Fig. 6. Note the image being scanned and the plug inserted into the joystick jack.

### Developing Driver Software

It's surprisingly easy to develop driver software for the digitizer. First, it's necessary to open a communications file that permits Junior to send instructions

to the plotter. If you use the HP7470, insert this line near the beginning of your programs: OPEN "COM1:2400,S,7,1,RS,CS65535,DS,CD" AS #1. For more information about modifying this plotter communications protocol, perhaps for use with other plotters, see Junior's documentation, especially the explanation of the OPEN "COM..." statement on


### LISTING 1. AUTOMATIC DIGITIZER PROGRAM FOR PCjr AND HP7470 XY PLOTTER.

```

10 'AUTOMATIC OPTICAL DIGITIZER
FOR
20 'PCjr AND HP7470 XY PLOTTER
30 'COPYRIGHT 1984 BY FORREST M.
MIMS
40 KEY OFF:CLS:SCREEN 1,0:COLOR 1
50 SENSOR=STICK(0)
60 LOCATE
15,15:PRINT"SENSOR="SENSOR
70 LOCATE 1,1:PRINT"CALIBRATION
PROCEDURE;"
80 LOCATE 3,1:PRINT"ADJUST
SYSTEM TO PROVIDE A
DIFFERENCE OF"
90 LOCATE 4,1:PRINT"AT LEAST
TWO UNITS WHEN PHOTOCCELL
VIEWS"
100 LOCATE 5,1:PRINT"A LIGHT AND
DARK BACKGROUND;"
110 LOCATE 7,1:PRINT"PRESS ANY
KEY WHEN READY;"
120 RS=INKEY$:IF RS="" THEN 50
130 CLS:LOCATE 1,1:PRINT"NEXT,
SUBTRACT 1 FROM THE HIGHER
VALUE"
140 LOCATE 2,1:INPUT"AND ENTER
THE DIFFERENCE: ",N
150 'PCjr-HP7470 PLOTTER PROTOCOL
160 OPEN"COM1:2400,S,7,1,RS,CS65535,
DS,CD" AS #1
170 SCREEN 3
180 PRINT #1,"SC0,159,0,199;"
190 FOR X=0 TO 159 STEP 2
200 FOR Y=0 TO 199 STEP 2
210 PRINT #1,"PA",X,Y;"
220 SENSOR=STICK(0)
230 LOCATE
1,1:PRINT"SENSOR="SENSOR
240 R=159-X
250 IF SENSOR > N THEN PSET (R,Y)
260 IF Y < 2 THEN GOSUB 310
270 NEXT Y
280 NEXT X
290 LOCATE 1,1:PRINT"
"
300 GOTO 300
310 FOR A=1 TO 200:NEXT A
320 RETURN

```

**Fig. 7. Automatic digitizer scan pattern**



pp. 4-240-246 of IBM's PCjr BASIC manual.

Next, it's necessary to develop a simple routine that scans the plotter's pen carriage across the paper being digitized. The HP7470 and other plotters can be easily scaled to match Junior's screen resolution. In the low-resolution mode (Screen 3), Junior's screen is divided into a grid consisting of 160 vertical and 200

## Computer Scientist




Fig. 9. Screen photographs of test shapes in Fig. 8.

horizontal boxes. This statement assigns the same scale to the HP7470: PRINT #1 "SC0,159,0,199;"

Now a simple BASIC routine can be developed to scan the pen carriage across the paper. The scan pattern can permit the fiber optic sensor to illuminate and examine each of  $160 \times 200$  imaginary boxes on the paper being digitized. Or the program can be easily modified for faster operation by having the sensor skip any specified number of boxes.

### Caution.

Never include pen selection or replacement instructions in your programs! To do so might damage the pen carriage mechanism. The only exception to this rule is if you use a detector housed in a pen and connected to the computer with flexible wires that do not interface with movements of the carriage or tangle with the pen stalls.

Listing 1 is a sample program that will help you understand how to develop driver software for your system. Here's how it works:

Line 50 assigns to a register labeled SENSOR the numerical value of the photocell reading [STICK(0)]. Lines 60-140 provide a calibration procedure that allows you to input a number that enables

the computer to determine when the sensor is viewing a light or dark background. This number is determined by moving the pen carriage via the plotter's front panel controls across dark and light patterns on a paper while observing the readout window appearing on the monitor screen.

Lines 150-180 establish the communications protocol for the computer and the plotter. Lines 190-280 scan the pen carriage across the paper and paint a replica of the image being scanned on the computer's monitor. The scan resolution is determined by the Step values in lines 190 and 200. The values given in Listing 1 (2) can be changed to give higher or lower resolution.

The subroutine (lines 310-320) was added to provide a brief delay that prevents the pen carriage from making unwanted back and forth movements at the beginning of each scan. These movements sometimes cause false data to be plotted on the monitor screen. If you use a different plotter, you may wish to modify or eliminate this subroutine.

Finally, line 290 erases the readout window that provides the on-screen sensor readout when an image is being digitized. You may wish to omit this line and line 230 if the readout window blocks part of the image.

Figure 7 shows the scan pattern that results when the scan increment steps in lines 190 and 200 are increased to 10. You may wish to modify the scan pattern to provide bidirectional sensing. This will speed up the digitizing process.

Figure 8 shows two test shapes drawn with a black marker pen on ordinary typing bond and run through the automatic digitizer. The resultant screen images are shown in Fig. 9.

Note that the aspect ratio of the shapes has been altered. For example, the circular smile face appears as an oval in the screen photo in Fig. 9. This occurred because I manually altered the origins of the plotter (P1 and P2) so that the pen carriage scanned only a window containing the image and not the surrounding white space. Of course it's possible to preserve the proper aspect ratio by exercising the proper care when setting up the scanning window.

Incidentally, note that the triangle in Fig. 8 has no side parallel to the axes of the paper upon which it was drawn. I purposely avoided having a parallel side to eliminate any possibility of missing one of the triangle's sides during the scan. This can occur, as I found out the hard way, should a line parallel to the vertical axis fall halfway between two passes of the moving pen carriage.

### Going Further

The basic techniques described thus far can be easily modified and expanded. For example, it's possible to devise relatively simple software that examines blacked-in answer boxes on a quiz sheet and tabulates test scores. By opening a second file, it's possible to store digitized images in Junior's RAM and on disk. Images can include maps, surveys, diagrams, graphics, charts, and possibly high-contrast photographs.

Though my digitizer experiments have thus far been limited to the PCjr and the HP7470, with suitable modifications many other combinations of computers and plotters should also work. The Color Computer should be particularly well suited.

You may also wish to explore other sensing methods. Figure 10, for example, is a straightforward phototransistor sensing circuit whose output has a potential of 5 V when the phototransistor is dark. Otherwise, the output is at ground (0 V). This kind of circuit can be easily adapted for computers that have voltage-dependent joystick ports.

For more information about light sensing circuitry, see *The Forrest Mims Circuit Scrapbook* (McGraw-Hill, 1983), pp. 43-45. ◇

Fig. 10. Simple phototransistor sensor circuit.

